Диалог с пользователем

*       *       *

Продолжим изучение программирования на Visual Basic for Applications. Подробные описания теперь уже не нужны - достаточно лишь заострения внимания на некоторых "подводных камнях" или скрытых возможностях.

В этой главе речь пойдет о так называемых "пользовательских формах" - формально они являются "нестандартными диалоговыми окнами" программ, наподобие диалоговых окон Word, но по сути с их помощью можно создавать настоящие программы, подобные компилируемым программам для Windows, которые будут отличаться от них лишь тем, что для их работы необходим Microsoft Word. К сожалению, их нельзя создавать с помощью средства записи макросов, но зато все средства создания форм снабжены подробной справкой.
Как и прежде, в качестве основной среды разработки будет использоваться Microsoft Office 97, об отличиях же более старших версий Microsoft Office будет упоминаться тогда, когда это имеет значение.
Для создания формы в Редакторе VBA необходимо из контекстного меню правой кнопки мыши в Менеджере проектов выбрать "Вставить - UserForm" (рис. 6.1).

Рис. 6.1. Окно дизайна форм.

Более подробно о формах, их элементах и свойствах рассказывалось в четвертой главе книги, поэтому ограничусь кратким повторением.

Форма состоит из элементов - различных средств ввода и отображения информации (командных кнопок, мест для ввода текста, мест для вывода текста, переключателей и флажков и др.), которые могут на ней располагаться. У каждого элемента есть свойства - ширина, высота, цвет, особенности внешнего вида и поведения. 

Основное отличие программного кода, обслуживающего формы, от кода в обычных модулях заключается в принципе программирования. В то время как программа на Visual Basic for Applications вроде той, что была нами совместно написана в предыдущей главе книги, выполняется "последовательно", - то есть все ее команды выполняются одна за другой, и их порядок выполнения изменяется только в зависимости от операторов условного и безусловного перехода "If" и "Goto" (в принципе - в частности могут быть отклонения от данного правила), форма "реагирует на события" - то есть каждому элементу формы может быть поставлена в соответствие определенная отдельная программа, выполняющая нужные команды при совершении с этим элементом каких-либо действий. Большинство программ для Windows, да и сам текстовый редактор Microsoft Word построены именно по такому принципу - они не "работают сами по себе", а ждут от пользователя команд и "реагируют" на них: выполняют те или иные действия. Так и форма: программист должен разработать ее интерфейс - создать форму, расположить на ней элементы, определить их свойства, а затем написать "программы обработки событий", поместив в каждой такой программе код, который должен быть выполнен, когда то или иное событие произойдет, то есть будет нажата какая-нибудь кнопка на форме или на клавиатуре, введен текст в поле ввода текста, выбран тот или иной флажок на форме и др. Программы обработки событий могут изменять и внешний вид самой формы и ее элементов: прятать их или показывать, изменять любые их свойства, так как у каждого элемента формы есть свое уникальное (т.е.принадлежащее только этому элементу и больше никакому другому) имя, по которому к нему можно обращаться. 
Для примера создадим форму, в которой будет производиться подсчет количества теплоты, выделяемой в проводнике при протекании в нем тока. Согласно соответствующим формулам, Q(количество теплоты в джоулях)=U2(напряжение в вольтах)*t(время в секундах)/R(сопротивление в омах), при этом R=p(удельное сопротивление материала проводника в омах на метр)*l(длина проводника в метрах)/s(площадь поперечного сечения проводника в квадратных сантиметрах). Таким образом, формула для расчета количества теплоты выглядит так (она также известна под именем закона Джоуля-Ленца): Q=(U2*t*s)/(l*p). Ее мы и запрограммируем в форме.

Рис. 6.2. Закон Джоуля-Ленца.

Экскурс в физику

В школьном курсе физики более часто используется формула Q=I2*R*t=(I2*p*l*t)/s (обозначения такие же: Q - количество теплоты в джоулях, I - сила тока в амперах, t - время в секундах, p - удельное сопротивление материала проводника в омах на метр, l - длина проводника в метрах, s - площадь поперечного сечения проводника в квадратных сантиметрах). Однако в реальности мы значительно чаще имеем дело с электрическими системами, в которых измерению и контролируемому изменению поддается именно напряжение, а не сила тока. 
Не удивляйтесь, что согласно данной формуле выделение тепла окажется максимальным при большой толщине проводника и его малой длине, хотя, казалось бы, должно быть наоборот. Дело в том, что при большой толщине проводника и его малой длине (и тем самым малом сопротивлении) в условиях постоянного напряжения резко увеличивается сила тока в цепи, которая и разогревает этот самый проводник. Эксперимент, подтверждающий правильность данной формулы, может проделать каждый: если электрическую лампочку встряхнуть так, чтобы ее нить пересеклась сама с собой (тем самым уменьшится длина нити, через которую проходит ток), то нить будет светиться гораздо ярче и греть сильнее. Если уменьшать длину проводника до минимума (в условиях того же постоянного напряжения), то тепловыделение возрастет настолько, что разрушит проводник, расплавив его - возникнет ситуация короткого замыкания. Поэтому при применении вышеуказанной формулы для расчета нагревательного элемента следует учитывать рост силы тока при уменьшении длины проводника или увеличении его диаметра и сопоставлять величину этой силы с возможностями электрической цепи (обычно они ограничены свойствами предохранителя, выражаемыми в просторечии словами типа "пробка на 20 ампер"). 
Формула, рассматриваемая в данной главе, может быть использована, к примеру, при оценке способности подаваемого на линию напряжения в течение определенного времени приводить к очистке ото льда обледеневшего провода путем его разогревания, - ситуация, встречающаяся зимой на линиях электропередач. В том случае, если выделяемого тепла недостаточно для растапливания льда, налипшего по всей длине провода, следует увеличивать напряжение на проводе, что и делается на ЛЭП (штатные потребители при этом отключаются).
Сначала создадим саму форму и разработаем ее дизайн. Как нетрудно сообразить, в ней должно быть пять полей для ввода значений - напряжение, время, диаметр проводника, его длина и удельное сопротивление материала, из которого он сделан, одно поле для вывода значения - количества теплоты, и кнопка выхода из программы. Но мы добавим еще кнопку - "Вставить значение в документ": пусть при нажатии этой кнопки подсчитанная информация будет вставлена в текст активного документа. Не мешало бы еще также поместить на форме текст о назначении данной программы и краткую инструкцию по ее использованию, а также сделать у формы приличный заголовок (свойство "Caption" элемента "UserForm").

Для помещения на форму элемента управления достаточно перетащить его с "Панели элементов" на форму. 

Вот что может у вас получиться - рис. 6.3.

06_03.bmp
Рис. 6.3. Форма нашей программы.

Поля ввода параметров имеют имена "TextBox1"…"TextBox5" соответственно (такие имена по умолчанию задаются автоматически при создании полей), поле отображения результата имеет имя "TextBox6", кнопки имеют имена "CommandButton1" и "CommandButton2" сверху вниз соответственно. В элементы "TextBox1"…"TextBox5" пользователь будет вводить текст. Желательно установить свойство Locked элемента "TextBox6" как "True" (рис. 6.4), чтобы не допустить случайного ввода пользователем в него текста - к сбою это не приведет, но вызовет неудобство для пользователя, так как при подсчете результата этот текст придется удалить.

Рис. 6.4. Установка свойства Locked элемента "TextBox6" как "True".

Разработка дизайна программы - ответственный момент, но не менее важным является обдумывание принципов работы программы. Помните, что исправление ошибки на стадии проектировки программы в несколько раз легче ее исправления на стадии реализации, в десятки раз легче, чем на стадии распространения, и в сотни раз легче, чем на стадии внедрения. Можно сделать, например, так: пользователь вводит все значения, нажимает кнопку "Подсчитать" (надо будет добавить на форму…), и ему в окне результата он выдается. Но в этом случае пользователь вынужден выполнять лишнее действие - нажатие на кнопку. Кроме того, так как обязательно надо продумать систему защиты от неправильных действий пользователя (например, нельзя допускать ввод нулевых или нечисловых значений в поля "TextBox4" и "TextBox5"), то при использовании данного способа подсчета придется либо выдавать при неправильном вводе сообщение об этом, либо, что представляется более красивым, ставить в эти поля значения по умолчанию, как только пользователь сделает неправильный ввод и уберет курсор с поля ввода. Но… все же, как неудобно! Нажимать лишний раз кнопку, постоянно следить за тем, стоят ли в полях ввода именно нужные значения, а не значения по умолчанию (ведь можно при вводе и случайно задеть соседнюю клавишу), да и при программировании проблем будет предостаточно, - изволь предусмотреть все случаи, на каждый написать программу обработки именно этого случая… Не слишком ли много проблем? А может быть, надо просто получше подумать о проектировании программы?

Основной принцип такого проектирования - "сделай проще, но без ущерба функциям". Что нам надо? Чтобы был результат - отображался итог вычислений. Когда он может быть посчитан? Да когда определены все значения в полях ввода, и при этом два нижних значения (то есть "TextBox4" и "TextBox5") ненулевые. Так пусть результат в своем окне появляется тогда и только тогда, когда все эти условия выполнены. И пусть программа постоянно отслеживает ввод значений в поля ввода и, как только будут введены все пять значений, и последние два будут ненулевые, а остальные - числовые, отобразит результат. Но есть ли такая возможность у языка Visual Basic for Applications - отслеживать момнеты ввода данных в поля ввода текста? А проверим! Откроем окно программного кода формы (из контекстного меню правой кнопки мыши нашей формы в Менеджере проектов выберем "Программа" - рис. 6.5) и из выпадающего списка в левом верхнем углу (рис. 6.6) выберем, например, "TextBox1". 

Рис. 6.5. Так можно открыть окно программного кода формы.

Рис. 6.6. Выпадающий список в левом верхнем углу - навигатор по программам элементов формы.

Так… появился фрагмент кода:
Private Sub TextBox1_Change()

End Sub
"Change" - это по-английски "изменение". Код, написанный в этой части программы (часть эта называется "обработчик события "Change"), должен выполняться всякий раз при происхождении этого события. Можно предположить, что это событие происходит тогда, когда в поле ввода ввели какой-либо символ или удалили его оттуда. В этом случае это - именно то событие, которое нам надо. Но то ли? Проверим. Напишем-ка здесь строчку:

Private Sub TextBox1_Change()

TextBox6.Text=TextBox1.Text

End Sub

Пусть для эксперимента при изменении текста в первом поле ввода произойдет изменение текста в поле отображения результата. Проверим, будет ли это работать, - нажмем F5 (запуск программы на выполнение) и введем текст в первое поле ввода. 
Прекрасно - в поле отображения результата появляется тот же текст! Значит, событие "Change" - именно то, которое нам надо использовать в нашей программе, чтобы после каждого нового ввода данных пользователем проверять условия возможности отображения результата и отображать его в случае их выполнения. Ну, а если бы оно нам не подошло, то пришлось бы перебирать все остальные события из выпадающего списка вверху справа окна программного кода формы и для каждого проводить такое же исследование: помещать в обработчик этого события команду, выполняющую какое-либо действие и смотреть, будет ли она выполнена при вводе текста в первое поле ввода.

Проверка возможности подсчета результата и вывод его в случае соблюдения описанных выше условий должна происходить после каждого ввода какого-либо символа в любое из окон ввода, а также при удалении символа из любого окна ввода. Писать пять одинаковых программ для каждого из окон ввода представляется нецелесообразным, поэтому лучше проверку возможности отображения результата и его подсчет вынести в отдельную подпрограмму-процедуру, а из каждого обработчика события "Change" полей ввода ее вызывать.

Переведем на язык Visual Basic for Applications условие возможности отображения результата. Во-первых, все значения полей ввода должны быть числовыми. В Visual Basic for Applications есть специальная функция для проверки того, является ли та или иная строка записью числа - IsNumeric Можно, например, найти ее описание в разделе "Поиск" Справочной системы по VBA из Microsoft Office 97 по словам "строковое выражение числовое значение" - рис. 6.7. 

06_07.bmp
Рис. 6.7. Справка по функции IsNumeric в Microsoft Office 97.

(В Microsoft Office 2000/XP информацию об этой функции так просто не получить, однако можно попробовать задать вопрос Мастеру ответов, например, так: "Does the expression a number?" - "Является ли это выражение числом?" В этом случае в перечне результатов поиска будет и справка по функции IsNumeric (рис.6.8).

Рис. 6.8. Справка по функции IsNumeric в Microsoft Office XP.

Однако найти ее, не зная имени функции, все же непросто - при другой формулировке запроса эта информация может вообще не быть выведена даже в перечне результатов поиска.)
 Ее (функцию IsNumeric) и используем. Ну, а для проверки отличия от нуля значений в последних двух полях ввода используем функцию "Val", которая переводит строковое выражение в числовое, если это строковое выражение содержит в своем начале цифры (или все состоит из них).
Итак, результат существует, если 
IsNumeric(TextBox1.Text) = True And IsNumeric(TextBox2.Text) = True And IsNumeric(TextBox3.Text) = True And IsNumeric(TextBox4.Text) = True And IsNumeric(TextBox5.Text) = True And Not Val(TextBox4.Text) = 0 And Not Val(TextBox5.Text) = 0
В этом случае можно осуществить подсчет по формуле:
rez = ((Val(TextBox1.Text) ^ 2) * Val(TextBox2.Text) * Val(TextBox3.Text)) / (Val(TextBox4.Text) * Val(TextBox5.Text))
и отобразить его в поле вывода результата:
TextBox6.Text = Str$(rez)
Функция Str делает преобразование, обратное тому, что производит Val - конвертирует числовое значение выражения в строковое, что позволяет этому значению в дальнейшем обрабатываться как строке.
(Функции Val и Str нужны "для самого VBA" - они, не производя никакого видимого действия, преобразуют ни что иное, как типы данных обрабатываемых ими переменных, - соответственно из строкового в числовой и из числового в строковый. Подробнее о типах данных - в четвертой главе.)
Теперь можно написать процедуру вычисления результата и вызовы ее из всех обработчиков событий "Change":
Private Sub TextBox1_Change()

Scet

End Sub

Private Sub TextBox2_Change()

Scet

End Sub

Private Sub TextBox3_Change()

Scet

End Sub

Private Sub TextBox4_Change()

Scet

End Sub

Private Sub TextBox5_Change()

Scet

End Sub

Private Sub Scet()

If IsNumeric(TextBox1.Text) = True And IsNumeric(TextBox2.Text) = True And IsNumeric(TextBox3.Text) = True And IsNumeric(TextBox4.Text) = True And IsNumeric(TextBox5.Text) = True And Not Val(TextBox4.Text) = 0 And Not Val(TextBox5.Text) = 0 Then

rez = ((Val(TextBox1.Text) ^ 2) * Val(TextBox2.Text) * Val(TextBox3.Text)) / (Val(TextBox4.Text) * Val(TextBox5.Text))

TextBox6.Text = Str$(rez)

Else

TextBox6.Text = ""

End If

End Sub
В принципе программа уже почти закончена, но стоит еще разобраться с командными кнопками. Для кнопки "Отмена" обработчик события "Click" (то есть нажатия на кнопку) прост - выход из программы и выгрузка формы из памяти:
Private Sub CommandButton2_Click()

Unload Me

End Sub
Но у нас есть еще вторая кнопка - "Вставить результат в документ". А пусть она вставит в документ не просто значение результата, а фразу, содержащую как результат, так и введенные параметры! Это можно сделать командой
Selection.Text = "При прохождении тока напряжением в " + TextBox1.Text + " вольт по проводнику длиной " + TextBox4.Text + " метров, сечением " + TextBox3.Text + " кв.мм и удельным сопротивлением " + TextBox5.Text + " ом на метр за " + TextBox2.Text + " секунд выделится" + TextBox6.Text + " джоулей теплоты. "
Она сформирует фразу из значений полей ввода и вставит ее в активный документ. Проверим… так, действительно вставляет, но фраза остается выделенной. А значит, что следующая фраза, выведенная с помощью нашей программы, сотрет предыдущую. Надо посмотреть, нет ли в Visual Basic for Applications функции снятия выделения? Посмотрим по выпадающему меню после точки "Selection."… есть команда "Collapse" (то есть "Свернуть") - рис. 6.9. 

Рис. 6.9. Список методов обьекта Selection.

Из справки по ней (рис. 6.10) узнаем ее синтаксис:

Selection.Collapse Direction:=wdCollapseEnd.

Рис. 6.10. Справка по команде Collapse.

Эта команда убирает выделение и помещает курсор в его конец.

Можно также вставить текст в активный документ "Selection.TypeText Text:="Мой текст" (двоеточие после слова "Text" обязательно). Тогда в специальном снятии выделения со вставленного текста не будет необходимости, - оно будет автоматически сворачиваться к своему концу.

А если вдруг пользователь вызовет программу тогда, когда открытых документов в Word'е нет? Тогда ведь возникнет ошибка программы. Но этого легко избежать - надо просто проверить перед вставкой, есть ли открытые документы и, если нет, создать новый:
If Documents.Count = 0 Then Documents.Add
Осталась еще одна маленькая деталь. Кнопка "Вставить результат в документ" не должна работать, если результат вычислить нельзя (то есть поле "TextBox6" пусто). Как это сделать? Среди всего набора возможных свойств элемента "CommandButton" есть свойство "Enabled" (рис. 6.11) - если его установить как "False" (то есть "ложно"), то кнопка будет отображаться серым цветом и не будет реагировать на события (станет неактивной). 

Рис. 6.11. Свойство Enabled элемента CommandButton.

Это свойство можно задать и программно - командой "CommandButton1.Enabled = False". Поставим в процедуру вычисления результата пару команд, активизирующих кнопку, когда результат вычисляется и вставка его в текст возможна, и инактивирующих ее в противном случае, а также зададим с самого начала в форме свойство "Enabled" этой кнопки как "False", - чтобы она была неактивной с самого начала, до того, как в какое-либо окно ввода будут вводиться символы и начнет срабатывать процедура вычисления результата с командами задания состояния кнопки. 

Вот готовый код нашей программы:
Private Sub CommandButton1_Click()

If Documents.Count = 0 Then Documents.Add

Selection.Text = "При прохождении тока напряжением в " + TextBox1.Text + " вольт по проводнику длиной " + TextBox4.Text + " метров, сечением " + TextBox3.Text + " кв.мм и удельным сопротивлением " + TextBox5.Text + " ом на метр за " + TextBox2.Text + " секунд выделится" + TextBox6.Text + " джоулей теплоты. "

Selection.Collapse Direction:=wdCollapseEnd

End Sub

Private Sub CommandButton2_Click()

Unload Me

End Sub

Private Sub TextBox1_Change()

Scet

End Sub

Private Sub TextBox2_Change()

Scet

End Sub

Private Sub TextBox3_Change()

Scet

End Sub

Private Sub TextBox4_Change()

Scet

End Sub

Private Sub TextBox5_Change()

Scet

End Sub

Private Sub Scet()

If IsNumeric(TextBox1.Text) = True And IsNumeric(TextBox2.Text) = True And IsNumeric(TextBox3.Text) = True And IsNumeric(TextBox4.Text) = True And IsNumeric(TextBox5.Text) = True And Not Val(TextBox4.Text) = 0 And Not Val(TextBox5.Text) = 0 Then

rez = ((Val(TextBox1.Text) ^ 2) * Val(TextBox2.Text) * Val(TextBox3.Text)) / (Val(TextBox4.Text) * Val(TextBox5.Text))

TextBox6.Text = Str$(rez)

CommandButton1.Enabled = True

Else

TextBox6.Text = ""

CommandButton1.Enabled = False

End If

End Sub
Назначить форме кнопку или пункт меню для вызова из Word нельзя, - это возможно только для модулей. Поэтому переименуем для красоты форму в "Teplotok", например (свойство "Name" обьекта "UserForm" - можно задать в окне свойств, выделив форму) и напишем модуль, в котором будет всего одна команда - вызов созданной нами формы:
Sub TeploCount()

Teplotok.Show

End Sub
Зададим модулю красивое имя ("Teplo", например, через свойство "Name" обьекта "Модуль1" (рис. 6.12), если модуль для программы вызова формы был вставлен с помощью пункта "Вставить - Модуль" контекстного меню правой кнопки мыши в Менеджере проектов) и назначим в Word кнопку для вызова макроса "Normal.Teplo.TeploCount". Вот и все - наша программа готова (рис. 6.13).

Рис. 6.12. Задание модулю нужного имени.

Рис. 6.13. Окно готовой программы.

Теперь ее можно вызывать на выполнение соответствующей кнопкой. При желании можно скопировать форму и модуль в отдельный шаблон и создать в нем панель инструментов с кнопкой вызова макроса, - тогда программу можно будет установить и на другие компьютеры, скопировав шаблон с макросом в папку автозагружаемых файлов Word.

